

## Asservissement de puissance dans une résistance

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDE](#)

Dans l'automatique enseignée dans le supérieur, l'asservissement de puissance est moins connu que les asservissements de tension, courant, vitesse etc. Nous allons montrer dans ce document, la particularité du pilotage d'une puissance électrique.

### 1) Asservissement de puissance en continu.

#### 1.a Description

##### Schéma fonctionnel de principe

L'entrée de consigne, une tension  $V_e$ , indique la valeur que doit prendre la puissance  $P$ .

Comme tout système bouclé, il faut une information de retour qui indique la valeur de cette sortie : c'est le rôle du **capteur de puissance**, qui délivre une tension  $V_r$ .

L'asservissement agit pour que  $V_r$  suive  $V_e$ . Idéalement,  $V_r = V_e$ . Ceci est d'autant plus réalisé que la boucle possède une forte amplification.

Le comparateur est suivi d'un amplificateur de tension, capable de débiter dans  $R$ , de coefficient  $A$ , qui alimente la résistance  $R$ . Cette dernière dissipe la puissance  $P$ .




Schéma fonctionnel de principe

##### Schéma fonctionnel en schéma bloc

Le système bouclé peut se représenter par le formalisme des schémas blocs.

L'équation de la puissance,  $P = \frac{U^2}{R}$ , montre clairement que le système

est **non linéaire**.




Schéma fonctionnel en statique

**Remarque** : la grandeur  $U^2$  est forcément positive, et donc  $V_r$  également. Dans cet asservissement,  $V_r$  est inférieur à  $V_e$ . La tension  $V_e$  ne peut donc qu'être positive. Une tension  $V_e$  négative ne permettrait pas un point de fonctionnement stable. Le raisonnement s'applique également sur  $V_e$  qui ne peut pas changer de signe, sous peine de quitter un point de fonctionnement stable.

##### Schéma électronique simulé. **Fichier RI.cir**

Nous allons réaliser une simulation avec 2 sources de tension commandées :

- pour modéliser le comparateur qui délivre  $V_e - V_r$ , équipé d'un coefficient  $A$ , ce qui forme la chaîne directe,
- pour modéliser le capteur de puissance, par une source commandée par  $U$ , et qui délivre une tension de valeur  $\beta \frac{U^2}{R}$ , ce qui forme la chaîne de retour.

Ainsi connecté, le montage n'a que 3 nœuds, repérés (C), (2), (3).




Schéma simulé

#### 1.b Prédétermination du point de fonctionnement en continu.

- Choisissons un point de repos défini par une puissance  $P_0 = 2 \text{ W}$ , dissipée dans  $R = 50 \Omega$ .  
Il vient  $U_0^2 = P_0 \times R = 2 \times 50 = 100$ , d'où  $U_0 = \sqrt{100} = 10 \text{ V}$  (et  $I_0 = 20 \text{ mA}$ ).

On choisit une amplification  $A = 20$ . On a :  $V_\epsilon = \frac{U_0}{A} = \frac{10}{20} = 0,5 \text{ V}$ .

Le capteur a une transmittance  $\beta = 0,1 \text{ V/W}$ .

Le retour donnant  $V_r = \beta \frac{U^2}{R} = 0,1 \times \frac{100}{50} = 0,2 \text{ V}$ , il faudra donc une entrée de consigne de  $V_\epsilon + V_r = 0,5 + 0,2 = 0,7 \text{ V}$ .

- Si on veut disposer de la puissance à partir de  $V_e$ , il nous faut écrire le système :

- Nous avons 3 équations :
- (1)  $V_\epsilon = V_e - V_r$ , comparateur
  - (2)  $V_r = \beta P$ , chaîne de retour
  - (3)  $P = (A V_\epsilon)^2 / R$ , chaîne directe

Par (3) :  $(V_\epsilon)^2 = \frac{P R}{A^2}$

(1) et (2) :  $V_\epsilon = V_e - \beta P$  d'où  $(V_\epsilon)^2 = (V_e - \beta P)^2 = V_e^2 - 2 V_e \beta P + (\beta P)^2$

En combinant avec (3) :  $\frac{P R}{A^2} = V_e^2 - 2 V_e \beta P + (\beta P)^2$ , on aboutit à un trinôme :

$$\boxed{\beta^2 P^2 - P (2 \beta V_e + R/A^2) + V_e^2 = 0}$$

Reprenons notre application numérique :  $A = 20$ ,  $R = 50 \Omega$ ,  $\beta = 0,1 \text{ V/W}$ . Il vient :  $0,01 P^2 - P (0,2 V_e + 0,125) + V_e^2 = 0$ .

Vérifions : pour  $V_e = 0,7 \text{ V}$  :  $0,01 P^2 - 0,265 P + 0,49 = 0$ , qui admet 2 racines réelles : 2 W et une autre valeur (24,5 W) qu'on élimine.

On confirme ainsi la valeur de  $V_r$  :  $= \beta P = 0,1 \times 2 = 0,2 \text{ V}$


Autre point, pour  $V_e = 0,75 \text{ V}$  :  $0,01 P^2 - 0,275 P + 0,5625 = 0$ , qui admet 2 racines réelles : 2,2256 W et une autre valeur (25,27 W) qu'on élimine.

### 1.c Ecart autour du point de repos. Linéarisation

Prenons le cas d'un léger décalage de la tension de consigne, qui occasionne une nouvelle valeur en U, et donc un écart en  $U^2$ .

Si cet écart est de faible amplitude par rapport à  $U_0^2$ , on peut linéariser la fonction  $U^2$ , par sa dérivée au point de repos, soit  $2 U_0 u$ , où **u est l'écart**, et  $2 U_0$  une valeur numérique.

Il vient un **schéma fonctionnel linéarisé** :



#### Remarque :

on passe en lettres minuscules (v, u, p) pour signifier un écart autour du point de repos (V, U<sub>0</sub>, P<sub>0</sub>).

Si on reprend l'application numérique précédente, nous aurons :

- une chaîne directe qui vaut :  $\frac{A 2 U_0}{R} = \frac{20 \times 2 \times 10}{50} = 8$ ,

- et une fonction de transfert en boucle fermée :  $\frac{P}{v_e} = \frac{8}{1 + 8 \times 0,1} = 4,44$ .

Ce qui signifie qu'un écart  $v_e$  (au voisinage de  $V_e = 0,7 \text{ V}$ ) se traduira par une variation de puissance  $p = 4,44 v_e$  (au voisinage de  $P_0 = 2 \text{ W}$ ).

Par exemple, supposons un accroissement  $v_e$  de 0,05 V qui s'ajoute à 0,7 V, la puissance sera augmentée de  $4,44 \times 0,05 = 222 \text{ mW}$ , qui s'ajoute à 2 W et P passera donc à 2,22 W.


### 1.d Comportement à un échelon de faible amplitude

Nous allons à être amené, dans le paragraphe suivant, à placer un filtre passe bas du premier ordre dans la chaîne de retour. Observons la conséquence sur le plan dynamique.

Nous modifions le schéma pour que notre boucle soit équipée d'un circuit  $R_1 C_1$  placé dans la chaîne de retour. Le noeud 31 est ajouté.

$R_1 = 1 \text{ k}\Omega$   
 $C_1 = 0,33 \text{ }\mu\text{F}$   
 Posons  $\tau_1 = R_1 C_1$ .

**Fichier R2.cir**



***$R_1, C_1$  : filtre passe bas dans la chaîne de retour***

La réponse à l'échelon en  $v_e$  peut se déterminer en petits signaux, sur un système **linéaire**.


Exploitions les transformées de Laplace.

Du schéma fonctionnel en faible variation, on déduit sa TBF :

$$\frac{p(p)}{v_e(p)} = \frac{(2 A U_0 / R)}{1 + (2 A U_0 / R) \beta / (1 + \tau_1 p)}$$

expression qui peut se mettre sous la forme :  $K \frac{1 + \alpha \tau p}{1 + \tau p}$

avec  $K = \frac{2 A U_0}{R + 2 A U_0 \beta}$ ,  $\tau = R_1 C_1 \frac{R}{R + 2 A U_0 \beta}$  et  $\alpha = 1 + \frac{2 A U_0 \beta}{R}$



***Schéma fonctionnel en variations. La sortie est en Watt, le retour est en Volt***

L'**annexe 1** nous indique, pour une entrée échelon unitaire, l'expression de la sortie  $p(t)$  est :

$$p(t) = K \left[ 1 + (\alpha - 1) \exp\left(-\frac{t}{\tau}\right) \right] u(t)$$

Si on reprend notre application numérique :

$$K = \frac{2 \times 20 \times 10}{50 + 2 \times 20 \times 10 \times 0,1} \approx 4,44,$$

$$\tau_1 = 0,33 \text{ ms et } \tau = 0,33 \text{ ms} \frac{50}{50 + 2 \times 20 \times 10 \times 0,1} \approx 183,3 \text{ }\mu\text{s}.$$


$$\text{et } \alpha = 1 + \frac{2 \times 20 \times 10 \times 0,1}{50} = 1,8$$

Lors d'un échelon d'amplitude  $v_e$ , on attend donc une variation de puissance d'équation :

$$p(t) = v_e \times 4,44 \times \left[ 1 + 0,8 \exp\left(-\frac{t}{\tau}\right) \right] u(t)$$

A  $t = 0$ , on est en présence d'un pic (car  $\alpha > 1$ ) qui vaut  $K \alpha v_e$ , soit  $\approx 4,44 \times 1,8 v_e = 8 v_e$

Après le régime transitoire, la sortie  $p(t)$  tend vers  $K v_e$  soit  $= 4,44 v_e$ .



***réponse à l'échelon à  $\alpha > 1$***


Exemple d'application numérique :

Pour un échelon  $v_e = 50 \text{ mV}$  (qui s'ajoute à  $V_e = 0,7 \text{ V}$ ), on aura un pic de  $0,4 \text{ W}$  qui s'ajoute à  $2 \text{ W}$ .  $p(t)$  sera une exponentielle décroissante partant de  $2,4 \text{ W}$  se dirigeant vers la valeur finale  $K v_e = 0,22 \text{ W}$  (qui s'ajoute à la valeur de  $P = 2 \text{ W}$ ) avec une constante de temps de  $183,3 \text{ }\mu\text{s}$ .

## 2) Simulations

### 2.a Simulation statique **fichier R1.cir**

On teste le montage par une analyse .DC : on fait varier la tension d'entrée continument entre 0 et 10 V.



*Puissance (en W) fonction de l'entrée  $V_e$  (en V).*


On affiche la puissance dans la résistance par W(R). La puissance dissipée dans R fonction de la tension de consigne  $V_e$  présente une équation du deuxième degré.

Remarque : comme prédéterminé au paragraphe 1b, le passage à 2 W est à  $V_e = 0,7$  V, et pour  $V_e = 0,75$  V, la puissance est à 2,2256 W

X Values	700.000m	750.000m
W(R)	2.0000	2.2256

### 2.b Simulation de la réponse à l'échelon **fichier R2.cir**

#### Entrée de consigne et Puissance de sortie



*Trace supérieure : Entrée de consigne (en V)  
Trace inférieure puissance (en W).*


#### Interprétation :

On retrouve la réponse prédéterminée précédemment : pour un échelon  $v_e = 50$  mV (qui s'ajoute à  $V_e = 0,7$  V), on a un pic à 0,4 W qui s'ajoute à 2 W.

On a une exponentielle décroissante partant de 2,4 W se dirigeant vers 2,2256 W avec une constante de temps de 183,3  $\mu$ s, identifiable par la méthode des 63 %.

La puissance a donc suivi la consigne, le régime transitoire a duré  $3 \tau = 560 \mu$ s et présente un dépassement.

## Allure de la tension aux bornes de R



**Trace supérieure : Tension aux bornes de R (V)**  
**Trace inférieure : son carré (V<sup>2</sup>).**

A  $t = 0$ , on est au point nominal :  $U_0 = 10 \text{ V}$ , et donc  $U_0^2 = 100 \text{ V}^2$ .

Hormis la pointe liée au régime transitoire expliquée précédemment ( $\alpha > 1$ ), on a une tension  $U$  qui se stabilise à  $10,55 \text{ V}$  et  $U^2$  à  $112 \text{ V}^2$  environ. L'écart est donc de  $12 \text{ V}^2$  ce qui est faible devant  $100 \text{ V}^2$ . La linéarisation est justifiée.

### Remarque :

L'erreur de position (l'écart entre  $v_e$  et  $v_r$  en régime établi), peut être réduit par une augmentation de l'amplification statique de la boucle ouverte :  $\frac{2 A U_0 \beta}{R}$ .

Dans notre cas, elle vaut  $\frac{2 \times 20 \times 10 \times 0,1}{50} = 0,8$ . On a donc une erreur statique importante.

Mais en contrepartie, une valeur élevée à  $\frac{2 A U_0 \beta}{R}$  augmente directement la valeur à  $\alpha$ , c'est à dire le pic de puissance qui apparait lors des transitions instantanées, ce qui peut être contraire à un cahier des charges. Il faut donc faire un compromis entre les différentes exigences.

## 3) Asservissement de puissance en sinusoïdal

### 3.a Description


#### Présentation :

Considérons maintenant le cas où la résistance  $R$  doit être alimentée par un signal alternatif. Ce dernier ne peut pas être appliqué à l'entrée de l'asservissement car il présente des alternances négatives.

Le principe est de disposer de 2 entrées :

- une,  $V_e(t) = V_{e_{max}} \sin \omega t$ , recevant le signal sinusoïdal dont la puissance est à amplifier,
- l'autre,  $V_C$ , présentée ci-contre par une tension continue, sert de réglage (ou de consigne) de puissance.

La boucle est équipée d'un multiplieur de tension de coefficient  $k_m$ . (En pratique, on lui adjoindrait un suiveur capable de débiter sur  $R$ ). La tension  $U$  (et donc la puissance  $U^2/R$ ) est la contribution de l'entrée sinusoïdale  $V_e$  d'une part, et de l'action de la consigne  $V_C$  d'autre part au travers la branche de retour.



**L'entrée sinusoïdale est  $V_e(t)$ .**  
**La sortie est la  $P$  dissipée dans  $R$ .**  
 **$V_C$  assure la gestion de la puissance.**

**Explication :**

En régime permanent,  $V_C$  est constant,  $V_E$  est constant, et la tension de sortie est :  $u(t) = K V_{e_{max}} \sin \omega t$ .

Il vient :  $u^2(t) = (K V_{e_{max}} \sin \omega t)^2$

D'où :  $p(t) = \frac{u^2(t)}{R} = (K V_{e_{max}})^2 \times \frac{1 - \cos 2 \omega t}{2} = \frac{(K V_{e_{max}})^2}{2} - \frac{(K V_{e_{max}})^2 \cos 2 \omega t}{2}$

Nous trouvons que la puissance se décompose en une composante continue et d'une variation sinusoïdale à la pulsation  $2 \omega$ .

C'est  $P$ , la puissance efficace =  $\frac{1}{T} \int_0^T p(t) dt$  qui doit être asservie. Il est alors nécessaire d'intercaler un

montage intégrateur pour disposer de  $P$ , à partir de  $p(t)$ . Ici, cet étage est réalisé par un filtre passe bas  $R_1 C_1$  qui prend la valeur moyenne et rejette la pulsation en  $2 \omega$ .

Notre signal d'entrée est à 10 kHz, un filtre tel que  $\frac{1}{2\pi R_1 C_1} \ll 20 \text{ kHz}$  peut convenir. En fixant

arbitrairement  $R_1 = 1 \text{ k}\Omega$ , on a  $C_1 \gg 8 \text{ nF}$ . En prenant  $C_1 = 3,3 \mu\text{F}$ , on respecte l'inégalité, la coupure est à 50 Hz environ. (Une valeur inférieure à  $C_1$  ne réduirait pas suffisamment la fréquence 20 kHz). La constante de temps de ce filtre est 3,3 ms. En sortie du filtre passe bas, le signal sera représentatif de la puissance efficace  $P$ , mais après un régime transitoire de 10 ms environ.


**Remarques :**

- La mesure de puissance peut également se faire par mesure de  $u(t)$ , mesure de  $i(t)$ , et par le produit  $p(t) = u(t) i(t)$ . Cette façon de faire autorise la simulation avec une impédance  $Z$ , au lieu d'être limité à une résistance  $R$ .

Le calcul aboutit alors à  $P = U I \cos \phi$ , avec  $U = K V_{e_{max}} / \sqrt{2}$  et  $I = I_{eff}$  qui passe dans l'impédance  $Z$  et  $\phi$  l'angle de  $\vec{I}$  vers  $\vec{U}$ .



- Sous Pspice, pour accéder, en cours de simulation, à un courant dans une branche, il est nécessaire de placer une source de tension nulle. Cette dernière a été ajoutée en série à  $Z$  afin de disposer de  $I$ , pour la source « H » (tension commandée par un courant) qui délivre une tension, image de  $I$ .




*ajout d'une source de tension nulle pour accéder au courant I*

**Avec une charge résistive R**, il vient alors le schéma de simulation ci-contre :

Un multiplieur, dont les entrées sont aux nœuds (8) et (3) assure le produit  $U \times I$ , affecté d'un coefficient  $\beta = 1$ . Le résultat est une tension en (4), image de la puissance instantanée : 1 W est représenté par 1 V.

La coupure à 50 Hz réalisée par le filtre  $R_1 C_1$  permet de délivrer en V(41) une tension constante, sans composante à 20 kHz (ou de très faible amplitude).

**Fichier R3.cir**



*Schéma simulé*

Le schéma étant différent, les valeurs numériques précédentes ne peuvent pas être conservées. L'amplitude de  $V_e$  intervient directement. On garde néanmoins la même résistance  $R = 50 \Omega$ , et sensiblement la même puissance dissipée. Afin de réduire l'erreur statique, on donne une valeur plus élevée à  $A$ , qui passe à 50.

## 4) Simulations

### 4.a Simulation statique **fichier R3.cir**

#### Simulation statique


On fixe  $V_{e\max} = 5\text{ V}$ , et, par la directive .STEP, on lance des runs pour des valeurs de  $V_C$  choisies selon la liste : 0 0.2 0.6 1 1.4 1.8 2 2.1 2.5 V

Une durée de run de 3 ms permet d'atteindre le régime établi.

Après chaque simulation, on affiche, par `AVGX(W(R),0,05 ms)` la puissance efficace dissipée dans R.

Les différentes valeurs, en fin de simulation, à  $t = 2,9\text{ ms}$ , sont reportées ci-contre :

X Values	2.9000m
AVGX(W(R),0.05m)	0.000
AVGX(W(R),0.05m)	179.198m
AVGX(W(R),0.05m)	563.564m
AVGX(W(R),0.05m)	953.015m
AVGX(W(R),0.05m)	1.3446
AVGX(W(R),0.05m)	1.7374
AVGX(W(R),0.05m)	1.9341
AVGX(W(R),0.05m)	2.0325
AVGX(W(R),0.05m)	2.4266




*courbe  $P = f(V_C)$  à  $V_{e\max} = 5\text{ V}$*

En première approximation, l'allure de la courbe donne une fonction quasi linéaire entre  $V_C$  et P.

**Remarque :** pour  $V_C = 2\text{ V}$  on a 1,9341 W et pour 2,1 V on a 2,0325 W

### 4.b Simulation dynamique : changement de consigne instantané

On fait une variation instantanée de  $V_C$  en y plaçant une source PULSE. Il a été choisi une variation de tension de consigne de 2 V à 2,1 V.



*De haut en bas : ordre de consigne 2 V à 2,1 V, puissance instantanée p(t), puissance efficace : de 1,9341W à 2,0325 W*

#### Interprétation

Sur la courbe de puissance, on a un dépassement qui se présente à chaque changement de consigne. C'est lié à la forme d'onde de p(t), elle-même tributaire des valeurs de courant, tension aux instants de commutation. Ici, le régime transitoire dure 200  $\mu\text{s}$  environ.

Les valeurs stabilisées sont ensuite conformes à l'essai statique.


X Values	900.000u	1.9000m
AVGX(W(R),0.05m)	1.9341	2.0325

*puissance efficace pour 2 V et 2,1 V de tension de consigne.*


### 4.b Simulation dynamique sur charge inductive changement de consigne instantané

On recommence la simulation précédente en plaçant une inductance  $L = 1\text{ mH}$  en série avec R. La mesure du courant se fait sur le dipôle RL, et donc tient compte du

déphasage. Rappelons que ce dernier est  $\text{artg} \frac{L\omega}{R}$  arrière.



On vérifie dans un premier temps, en régime établi, que la détermination de la puissance en V(41) est conforme à  $P = U I \cos \varphi$ . On fait un zoom à  $t = 900 \mu s$ , c'est à dire à  $V_C = 2 V$ .




Par position du curseur, on relève :  
 amplitude de la tension en V(3) = 22,626 V, amplitude du courant circulant dans le circuit R L série = 273,445 mA.  
 L'écart entre ces deux crêtes est de 13,34  $\mu s$ . Le déphasage correspondant est de 48 °.  
 Remarque : pour  $V_C = 2 V$  on a 1,934 W et pour 2,1 V on a 2,0325 W.

X Values	825.000u	838.359u
V(8)	179.296m	273.445m
V(3)	22.626	13.669

En supposant ces signaux sinusoïdaux, on déduit les valeurs efficaces : 16 V, 168 mA.  
 La puissance résultante est  $P = U I \cos \varphi = 16 \times 0,168 \times \cos 48 \approx 1,8 W$ .

La théorie donne un déphasage  $\text{artg} \frac{L\omega}{R} = \text{artg} \frac{1 m \times 2 \times \pi \times 10 k}{50} = \text{artg} 1,256 = 51,5 \text{ °}$ . On est très proche de la valeur identifiée sur des signaux quasi sinusoïdaux.

Observons maintenant sur une plus grande étendue, afin d'analyser le changement de puissance :



On retrouve sensiblement les mêmes valeurs de puissance sans la présence de L. V(41) présente une composante à 20 kHz, ce qui explique le faible écart, à 900  $\mu s$ , entre 1,8975 V. et la valeur attendue 1,8 V.


X Values	900.000u	1.9000m
AVGX(W(R),0.05m)	1.8918	1.9891
V(41)	1.8975	1.9951
V(C)	2.0000	2.1000



**4.c Simulation dynamique** test aux perturbations (**fichier R4.cir**)

Par simplicité dans ce test, on remplace une charge purement résistive. Pour vérifier la robustesse de l'asservissement, on fait intervenir une perturbation. Cette dernière peut être une modification de la résistance. Pour ce faire, on choisit de commuter instantanément la valeur de R en R/2. Cela reviendrait expérimentalement à placer instantanément une autre résistance identique en parallèle à R.

Pour simuler cet essai, on a choisi de remplacer la résistance R par une résistance commandée par une tension. La bibliothèque de Pspice propose un « Voltage-Controlled Switch ». Sa valeur est  $R_{ON}$  et  $R_{OFF}$ , respectivement quand on applique  $V_{ON}$  (ou plus) et  $V_{OFF}$  (ou moins) sur une tension de commande.



Le choix a été : si sa tension de commande est supérieure à 4V, la résistance  $R_{ON}$  est 25 Ω, si la tension de commande est inférieure à 1 V, la résistance  $R_{OFF}$  est 50 Ω.

On va donc observer comment réagit l'asservissement de puissance quand la charge passe instantanément de 50 Ω à 25 Ω.

La consigne reste inchangée à 2 V. On conserve  $V_e(t) = V_{e_{max}} \sin \omega t$ , avec  $V_{e_{max}} = 5 V$  et la fréquence à 10 kHz.





Schéma simulé



De haut en bas : valeur de R qui passe de 50 Ω à 25 Ω, puis revient à 50 Ω, Tension de retour, puissance efficace dissipée par la résistance

**Interprétation**

Si on ignore le régime transitoire qui se produit à  $t = 0$ , (non représenté ici), on retrouve la condition précédente :  $P = 1,9341 W$  sous 50 Ω. L'asservissement est opérationnel car  $V(41) = 1,9348 V$ , pour une tension de consigne à 2 V.

X Values	900.000u	1.9000m
-AVGX(W(EMULT),0.05m)	1.9341	1.9541
V(41)	1.9304	1.9545
V(3)/I(Sint)	50.000	25.000

A  $t = 1 ms$ , la résistance passe de 50 Ω à 25 Ω, (car  $V_{com}$  passe de 0 V à 5 V).

Le changement de résistance fait réagir la boucle d'asservissement : la chaîne de retour reste voisine de 2 V (l'erreur de position s'est même affaiblie). La puissance dissipée par la résistance, qui peut se mesurer par celle fournie par le multiplieur qui délivre V(3), est passée à 1,9541 W : l'écart de 20 mW ne représente que 1% alors que la résistance a chuté de 50 %.

On peut noter que le régime transitoire dure 200  $\mu$ s environ.

## **CONCLUSION**

Ce document a illustré la complexité d'un asservissement de puissance : les paragraphes 1) et 2) ont montré le côté non linéaire (l'évolution de puissance n'est pas proportionnelle à l'entrée de consigne, le régime transitoire est dépendant du point de repos, et donc de la puissance débitée dans R).

Les paragraphes 3) et 4) ont permis de vérifier, en signal alternatif, que l'on pouvait, par l'ajout d'un multiplieur, faire évoluer la puissance de sortie en fonction d'une entrée de consigne. Et, qu'avec des valeurs numériques adaptées (gain de boucle élevé), cet asservissement était robuste.

C'est cette conclusion qui permettra de traiter l'article suivant qui est l'asservissement de puissance d'une antenne de téléphone portable : la puissance émise doit suivre une loi de commande donnée par la tension de consigne.

## Annexe 1

Calcul de la réponse indicielle d'un système dont la transmittance est  $\frac{S(p)}{E(p)} = \frac{1 + \alpha \tau p}{1 + \tau p}$ .

L'entrée est un échelon, soit  $E(p) = \frac{1}{p}$ . Il vient alors  $S(p) = \frac{1}{p} \frac{1 + \alpha \tau p}{1 + \tau p} = \frac{1}{p(1 + \tau p)} + \frac{\alpha \tau}{1 + \tau p}$


Après décomposition en éléments simples du premier terme :  $S(p) = \frac{1}{p} - \frac{\tau}{1 + \tau p} + \frac{\alpha \tau}{1 + \tau p}$ .

Par mise en facteur :  $S(p) = \frac{1}{p} + (\alpha - 1) \frac{\tau}{1 + \tau p}$ .

D'où l'original, terme à terme :  $[1 + (\alpha - 1) \exp(-\frac{t}{\tau})] u(t)$ .

D'où le tracé, trace supérieure entrée échelon d'amplitude  $E_0$ , trace inférieure  $s(t)$  :

$$s(t) = E_0 [1 + (\alpha - 1) \exp(-t/\tau)] u(t)$$



Commentaires :

- la valeur finale est toujours  $E_0$ , le terme en exponentiel tendant vers 0,
- la discontinuité à  $t = 0^+$  est  $\alpha E_0$  : cela explique le pic présent si  $\alpha > 1$ ,
- on retrouve la constante de temps  $\tau$  dans l'exponentielle,
- si  $\alpha = 1$ , on retrouve  $s(t) = E_0$  pour  $t > 0$ , ce qui est évident.

## Annexe2 : fichier R1.cir

```

asservissement de puissance dans une résistance
* fichier R1.cir
Ve C 0 DC=0.7 ; entrée de consigne

.param A= 20 ; pour limiter erreur statique ; 1.5
Eeps 3 0 VALUE = { {A}*(V(C) - V(2)) } ; comparateur et ampli

.param R = 50
R 3 0 {R} ; résistance de charge

.param beta=0.1 ;
Eretour 2 0 VALUE = { {beta}*(PWR(V(3),2)/{R}) } ; beta x U^2 / R

.OP
.DC Ve 0 1 10m

.probe
.end

```

Netliste, prête à simuler

## Annexe3 : fichier R2.cir

```
asservissement de puissance dans une résistance
* fichier R2.cir
Ve C 0 PULSE (0.7 0.75 0.5m 1u 1u 2m 4m); entrée

.param A = 20
Eeps 3 0 VALUE = { {A}*(V(C) - V(2)) } ; comparateur et ampli

.param R = 50
R 3 0 {R} ; résistance de charge

.param beta = 0.1
EU2 31 0 VALUE = {{beta}*PWR(V(3),2)/{R} } ; beta x U^2 / R
R1 31 2 1k
C1 2 0 3.3u
.TRAN 1u 5m 0u 1u
.probe
.end
```

*Netliste, prête à simuler*

## Annexe4 : fichier R3.cir

```
asservissement de puissance dans une résistance
* fichier R3.cir

.param Vemax=5
Ve 1 0 SIN (0 {Vemax} 10k ); entrée

.param consigne=2 ; pour essai statique
VC C 0 DC={consigne}
+ PULSE (2 2.1 1m 1u 1u 1m 2m) ; pour essai dynamique

.param km=1
EMULT 3 0 POLY(2) (1,0) (2,0) 0 0 0 0 {km} ; delivre V(1) x V(2) x km

.param R = 50
R 3 32 {R} ; résistance de charge
L 32 31 1m

.param beta=1
Vbidon 31 0 DC=0 ; pour prélever le courant
HU 8 0 POLY(1) Vbidon 0 1 ; délivre U, image de I
EmultP 4 0 POLY(2) (3,0) (8,0) 0 0 0 0 {beta} ; délivre U I = P avec facteur beta

R1 4 41 1k ; conserve DC, coupe à 50 Hz
C1 41 0 3.3u ; 3 tau = 10 ms

.param A=40
Eeps 2 0 VALUE = { {A}*(V(C) - V(41)) } ; comparateur et ampli

*.STEP param consigne list 0 0.2 0.6 1 1.4 1.8 2 2.1 2.5 ; statique

.TRAN 0.1u 3m 0m 0.1u
.probe
.end
```

*Netliste, prête à simuler*

## Annexe5 : fichier R4.cir

```
asservissement de puissance dans une résistance
* fichier R4.cir

.param Vemax=5
Ve 1 0 SIN (0 {Vemax} 10k ); entrée

VC C 0 DC=2

.param km=1
EMULT 3 0 POLY(2) (1,0) (2,0) 0 0 0 0 {km} ; delivre V(1) x V(2) x km

Vcom com 0 PULSE (0 5 1m 1u 1u 1m 2m)
.Model int Vswitch (Ron=25 Roff=50 Von=4 Voff=1)
Sint 3 31 com 0 int ; R de 50 à 25 ohm

.param beta=1
Vbidon 31 0 DC=0 ; pour prélever le courant
HU 8 0 POLY(1) Vbidon 0 1 ; délivre U, image de I
EmultP 4 0 POLY(2) (3,0) (8,0) 0 0 0 0 {beta} ; délivre u x i .

R1 4 41 1k ; conserve DC, coupe à 48,2 Hz
C1 41 0 3.3u ; 3 tau = 10 ms

.param A=40
Eeps 2 0 VALUE = { {A}*(V(C) - V(41)) } ; comparateur et ampli

.TRAN 0.1u 3m 0m 0.1u
.probe
.end
```

*Netliste, prête à simuler*

articles 1 à 43 : sur le livre

**Tableau récapitulatif des articles PDF disponibles sur ce site**

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	<a href="#">présentation</a>	<a href="#">document PDF</a>
44	Exemples basiques et des exercices...	<a href="#">présentation</a>	<a href="#">document PDF</a>
45	Un exemple de circuit passif	<a href="#">présentation</a>	<a href="#">document PDF</a>
46	Un oscillateur Colpitts	<a href="#">présentation</a>	<a href="#">document PDF</a>
47	Compensation en fréquence des amplificateurs opérationnels	<a href="#">présentation</a>	<a href="#">document PDF</a>
48	Un amplificateur à transistors bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
49	Une bascule D Flip Flop CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
50	Une porte XOR à transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
51	Un VCO à 12 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
52	Une PLL à moins de 20 transistors MOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
53	Un oscillateur à résistance négative	<a href="#">présentation</a>	<a href="#">document PDF</a>
54	Une charge électronique	<a href="#">présentation</a>	<a href="#">document PDF</a>
55	Un amplificateur en classe C	<a href="#">présentation</a>	<a href="#">document PDF</a>
56	Le monostable 74 123	<a href="#">présentation</a>	<a href="#">document PDF</a>
57	Un amplificateur en classe D	<a href="#">présentation</a>	<a href="#">document PDF</a>
58	Le transformateur en linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
59	La loi d'ohm thermique	<a href="#">présentation</a>	<a href="#">document PDF</a>
60	Le transformateur en non linéaire	<a href="#">présentation</a>	<a href="#">document PDF</a>
61	Robustesse d'un oscillateur en anneau	<a href="#">présentation</a>	<a href="#">document PDF</a>
62	Une alimentation stabilisée	<a href="#">présentation</a>	<a href="#">document PDF</a>
63	Modélisation d'un haut-parleur	<a href="#">présentation</a>	<a href="#">document PDF</a>
64	Un synthétiseur de fréquence	<a href="#">présentation</a>	<a href="#">document PDF</a>
65	Un ampli audio de Sparkfun	<a href="#">présentation</a>	<a href="#">document PDF</a>
66	Simulation logique et analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
67	Un oscillateur à relaxation	<a href="#">présentation</a>	<a href="#">document PDF</a>
68	Lecteur de TAG RFID 125 kHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
69	Diagramme de l'œil avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
70	Un amplificateur hautes fréquences	<a href="#">présentation</a>	<a href="#">document PDF</a>
71	Une bizarrerie enfin expliquée...	<a href="#">présentation</a>	<a href="#">document PDF</a>
72	Comprendre le paramétrage de la FFT	<a href="#">présentation</a>	<a href="#">document PDF</a>
73	La relation de Bennett	<a href="#">présentation</a>	<a href="#">document PDF</a>
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	<a href="#">présentation</a>	<a href="#">document PDF</a>
75	Une horloge biphase sans recouvrement	<a href="#">présentation</a>	<a href="#">document PDF</a>
76	Quelques simulations sur la diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
77	Un ampli classe A, avec transformateur de sortie	<a href="#">présentation</a>	<a href="#">document PDF</a>
78	Des stimuli pour PSpice	<a href="#">présentation</a>	<a href="#">document PDF</a>
79	Simuler le TL431 : zener ajustable	<a href="#">présentation</a>	<a href="#">document PDF</a>
80	Un ADC flash	<a href="#">présentation</a>	<a href="#">document PDF</a>
81	Une chaîne d'acquisition : S&H, ADC, DAC	<a href="#">présentation</a>	<a href="#">document PDF</a>
82	Un amplificateur 50 MHz	<a href="#">présentation</a>	<a href="#">document PDF</a>
83	Un dérivateur non inverseur	<a href="#">présentation</a>	<a href="#">document PDF</a>
84	Un amplificateur bipolaire avec push pull CMOS	<a href="#">présentation</a>	<a href="#">document PDF</a>
85	Rôle des répéteurs logiques dans un circuit intégré	<a href="#">présentation</a>	<a href="#">document PDF</a>
86	Un driver logique CMOS pour charge 50 ohms	<a href="#">présentation</a>	<a href="#">document PDF</a>
87	Des triggers de Schmitt et des applications	<a href="#">présentation</a>	<a href="#">document PDF</a>
88	Un filtre gaussien analogique	<a href="#">présentation</a>	<a href="#">document PDF</a>
89	Un générateur de bruit rose	<a href="#">présentation</a>	<a href="#">document PDF</a>

90	Un anémomètre à fil chaud : simulation comportementale	<a href="#">présentation</a>	<a href="#">document PDF</a>
91	Un oscillateur à pont de Wien stabilisé par CTN	<a href="#">présentation</a>	<a href="#">document PDF</a>
92	L'emballement thermique d'une diode	<a href="#">présentation</a>	<a href="#">document PDF</a>
93	Les puissances dans un amplificateur	<a href="#">présentation</a>	<a href="#">document PDF</a>
94	Asservissement de puissance dans une résistance	<a href="#">présentation</a>	<a href="#">document PDF</a>
95	Asservissement de la puissance émise par une antenne radio	<a href="#">présentation</a>	<a href="#">document PDF</a>
96	Un driver de LED de puissance	<a href="#">présentation</a>	<a href="#">document PDF</a>
97	Exploiter Pspice pour simuler des filtres numériques	<a href="#">présentation</a>	<a href="#">document PDF</a>
98	Un filtre en cosinus surélevé avec Pspice	<a href="#">présentation</a>	<a href="#">document PDF</a>
99	Effet de la température sur un amplificateur en classe A	<a href="#">présentation</a>	<a href="#">document PDF</a>
100	Un amplificateur à transistors JFET et bipolaires	<a href="#">présentation</a>	<a href="#">document PDF</a>
<b>Supplément, hors article :</b>			
<b>mon cours « Electronique pour les communications numériques », <a href="#">polycopié couleur 201 pages en pdf</a></b>			

[retour à l'écran d'accueil de ce site](#)